

X04BFFP

NAG Parallel Library Routine Document

Note: before using this routine, please read the Users' Note for your implementation to check for implementation-dependent details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

1 Description

X04BFFP outputs a set of real matrices distributed on a two-dimensional logical processor grid. It writes to output units connected to the root processor.

Each matrix on a logical processor is nominally denoted by A . Each matrix A and the row dimension m and the column dimension n are local to each logical processor in the Library Grid. In general, each m and each n may be different on different logical processors. This routine does not assume that the local matrices A are part of any particular global data structure. It prints the local matrices in the row-major ordering of the grid.

In general, each matrix A can be printed on a different output unit NOUT defined on the logical processor which owns the matrix A but in practice NOUT will generally have the same value on all logical processors and the matrices will then be written to a single file.

Optionally a standard title which identifies the logical processor can be printed followed by a blank line before a matrix is printed. As an example, the title

```
Array from logical processor {5,3}
```

can be printed before the matrix on logical processor {5,3}. The matrices are output with a **maximum record length of 80**. Options are also available to print column numbers of the matrices.

If the root processor is not present on the current Library Grid then the logical processor {0,0} of the current two-dimensional grid takes over the role of the root processor.

2 Specification

```
SUBROUTINE X04BFFP(ICNTXT, NOUT, M, N, A, LDA, FORMAT, TITOP,
1                CNUMOP, ICOFF, W, LDW, IFAIL)
  INTEGER          ICNTXT, NOUT, M, N, LDA, ICOFF, LDW, IFAIL
  DOUBLE PRECISION A(LDA,*), W(LDW,*)
  CHARACTER*1      TITOP, CNUMOP
  CHARACTER*(*)    FORMAT
```

3 Usage

3.1 Definitions

The following definitions are used in describing the data distribution within this document:

- m – the number of rows of a matrix on a logical processor.
- n – the number of columns of a matrix on a logical processor.
- m_{\max} – the maximum value of m taken over all logical processors (which have a matrix to be printed) except the root processor.
- n_{\max} – the maximum value of n taken over all logical processors (which have a matrix to be printed) except the root processor.

3.2 Global and Local Arguments

The following global **input** arguments must have the same value on entry to the routine on each processor and the global **output** arguments will have the same value on exit from the routine on each processor:

Global input arguments: IFAIL

Global output arguments: IFAIL

Note: the workspace argument *W* and the input arguments *FORMAT*, *TITOP*, and *CNUMOP* are referenced only on the root processor.

3.3 Distribution Strategy

All real matrices are defined locally on each logical processor. The values of *m* and *n* (which define the dimensions of the matrices) do not have to be identical on every logical processor.

4 Arguments

- 1:** *ICNTXT* — INTEGER *Local Input*
On entry: the Library context, usually returned by a call to the Library Grid initialisation routine *Z01AAFP*.
- Note:** the value of *ICNTXT* **must not** be changed.
- 2:** *NOUT* — INTEGER *Local Input*
On entry: unit number for output. If *NOUT* is non-positive, no output of this matrix (including the standard title which specifies the processor) is done.
- 3:** *M* — INTEGER *Local Input*
On entry: *m*, the number of rows of the local matrix. If *m* is non-positive, no output of this matrix (including the standard title which specifies the processor) is done.
- 4:** *N* — INTEGER *Local Input*
On entry: *n*, the number of columns of the local matrix. If *n* is non-positive, no output of this matrix (including the standard title which specifies the processor) is done.
- 5:** *A(LDA,*)* — DOUBLE PRECISION array *Local Input*
Note: the size of the second dimension of the array *A* must be at least $\max(1, N)$.
On entry: the local matrix *A*.
- 6:** *LDA* — INTEGER *Local Input*
On entry: the size of the first dimension of the array *A* as declared in the (sub)program from which *X04BFFP* is called.
Constraint: $LDA \geq M$ if $M > 0$, $N > 0$ and $NOUT > 0$; otherwise $LDA \geq 1$.
- 7:** *FORMAT* — CHARACTER*(*) *Local Input*
On entry: a valid Fortran format code on the root processor. This may be any format code allowed on the root processor, whether it is standard Fortran or not. *FORMAT* is used to output elements of the matrices. It may or may not be enclosed in brackets. Examples of valid values for *FORMAT* are '(F11.4)', '1PE13.5', 'G14.5' .

In addition, there is a special code which forces *X04BFFP* to choose its own format code:

FORMAT = '*' means that *X04BFFP* will choose a format code such that numbers will be printed to as many significant digits as are necessary to distinguish between neighbouring machine numbers. Thus any two numbers that are stored with different internal representations should look different on output. Whether they do in fact look different will depend on the run-time library of the Fortran compiler in use.

FORMAT is not referenced on other logical processors.

Constraint: The character length of *FORMAT* must not be greater than 80.

- 8:** TITOP — CHARACTER*1 *Local Input*
On entry: if TITOP = 'Y', then the standard title which identifies the logical processor is printed before the corresponding matrix is printed. Otherwise the standard title is not printed.
 TITOP is referenced only by the root processor.
- 9:** CNUMOP — CHARACTER*1 *Local Input*
On entry: indicates the type of labelling to be applied to columns of the matrices.
 If CNUMOP = 'L' then the numbering of columns is specific to the matrix being printed. The columns are numbered starting at ICOFF + 1 and ending at ICOFF + n for each matrix.
 If CNUMOP = 'G' then the columns are numbered consecutively in the order they are printed. The columns are numbered starting at 1 and ending at n_{total} where n_{total} is the sum of all (positive) values of n .
 If CNUMOP \neq 'L' or 'G' then the columns are not numbered.
 CNUMOP is referenced only by the root processor.
- 10:** ICOFF — INTEGER *Local Input*
On entry: offset value for column numbering if CNUMOP = 'L'.
- 11:** W(LDW,*) — DOUBLE PRECISION array *Local Workspace*
Note: the size of the second dimension of the array W must be at least $\max(1, n_{\text{max}})$. This array is referenced only on the root processor.
- 12:** LDW — INTEGER *Local Input*
On entry: the size of the first dimension of the array W as declared in the (sub)program from which X04BFFP is called.
Constraint: On the root processor LDW must be at least $\max(1, m_{\text{max}})$.
- 13:** IFAIL — INTEGER *Global Input/Global Output*
 The NAG Parallel Library provides a mechanism, via the routine Z02EAFP, to reduce the amount of parameter validation performed by this routine. For a full description refer to the Z02 Chapter Introduction.
On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this argument (described in the Essential Introduction) the recommended values are:
 IFAIL = 0, if multigridding is **not** employed;
 IFAIL = -1, if multigridding is employed.
On exit: IFAIL = 0 (or -9999 if reduced error checking is enabled) unless the routine detects an error (see Section 5).

5 Errors and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output from the root processor (or processor {0,0} when the root processor is not available) on the current error message unit (as defined by X04AAF).

5.1 Full Error Checking Mode Only

IFAIL = -2000

The routine has been called with an invalid value of ICNTXT on one or more processors.

IFAIL = -1000

The logical processor grid and library mechanism (Library Grid) have not been correctly defined, see Z01AAF.

IFAIL = $-i$

On entry, the i th argument was invalid. This error occurred either because a global argument did not have the same value on all logical processors, or because its value on one or more processors was incorrect. An explanatory message distinguishes between these two cases.

5.2 Any Error Checking Mode

IFAIL = 1

On entry, variable FORMAT is more than 80 characters long.

IFAIL = 2

The code supplied in FORMAT cannot be used to output a number. FORMAT probably has too wide a field width or contains an illegal edit descriptor.

6 Further Comments

6.1 Algorithmic Detail

None.

6.2 Parallelism Detail

Each matrix is brought to the root processor and is then printed. This is performed sequentially using the row-major ordering of the grid.

7 References

None.

8 Example

To find the eigenvalues and eigenvectors of a 14 by 14 symmetric matrix A , using routine F02FQFP, and to print the eigenvalues and the eigenvectors on the root processor. The element a_{ij} of the matrix A is taken as $a_{ij} = \max(i, j) + 1$ if $i = j$ and $a_{ij} = \max(i, j)$ if $i \neq j$. The routine F01ZRFP is used to generate the matrix A on a 2×2 logical processor grid. The number of columns of the matrix A on each logical processor, N_x , is equal to 4 on logical processors $\{0,0\}$, $\{0,1\}$, and $\{1,0\}$. On the final logical processor $\{1,1\}$, $N_x = 2$. The number of eigenvectors on each logical processor is given by N_x .

The routine X04BFFP is used twice to print the eigenvalues and the eigenvectors. In the third call, X04BFFP prints the eigenvectors corresponding to positive eigenvalues. All results are printed to standard output.

See also Section 8 of the document for F02FQFP for a simple example of use.

8.1 Example Text

```
*      X04BFFP Example Program Text
*      NAG Parallel Library Release 3 Revised. NAG Copyright 1999.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER       (NOUT=6)
      INTEGER          N
      PARAMETER       (N=14)
      INTEGER          MG, NG
      PARAMETER       (MG=2,NG=2)
      INTEGER          LDA, TDA
      PARAMETER       (LDA=N+N+2, TDA=(N/(MG*NG))+2)
      CHARACTER*20     FORMT
```

```

PARAMETER      (FORMAT='F12.5')
DOUBLE PRECISION ZERO
PARAMETER      (ZERO=0.0D0)
*
.. Local Scalars ..
INTEGER        ICNTXT, ICOFF, IFAIL, J, JP, MP, NP, NX
LOGICAL        ROOT
CHARACTER      CNUMOP, TITOP
*
.. Local Arrays ..
DOUBLE PRECISION A(0:LDA-1,0:TDA-1)
*
.. External Functions ..
INTEGER        OFFSET
LOGICAL        Z01ACFP
EXTERNAL       OFFSET, Z01ACFP
*
.. External Subroutines ..
EXTERNAL       F01ZRFP, F02FQFP, GMATA, X04BFFP, Z01AAFP,
+             Z01ABFP
*
.. Executable Statements ..
ROOT = Z01ACFP()
IF (ROOT) THEN
    WRITE (NOUT,*) 'X04BFFP Example Program Results'
    WRITE (NOUT,*)
END IF
*
*   Define the 2D processor grid
*
MP = MG
NP = NG
IFAIL = 0
CALL Z01AAFP(ICNTXT,MP,NP,IFAIL)
*
IFAIL = 0
*
*   Generation of the matrix A
*
CALL F01ZRFP(ICNTXT,GMATA,N,N,A(1,1),LDA,NX,IFAIL)
*
IF (ROOT) THEN
    WRITE (NOUT,*) 'Matrix'
    WRITE (NOUT,*)
    TITOP = 'Y'
    CNUMOP = 'G'
END IF
ICOFF = 0
IFAIL = 0
*
CALL X04BFFP(ICNTXT,NOUT,N,NX,A(1,1),LDA,FORMAT,TITOP,CNUMOP,ICOFF,
+           A(N+1,1),LDA,IFAIL)
*
IFAIL = 0
*
*   Solution of the symmetric eigenvalue problem
*
CALL F02FQFP(ICNTXT,N,A,LDA,NX,IFAIL)
*
*   Print the eigenvalues on the root processor
*
IF (ROOT) THEN
    WRITE (NOUT,*) 'Eigenvalues'

```

```

        WRITE (NOUT,*)
          TITOP = 'N'
      END IF
      ICOFF = 0
      IFAIL = 0
*
      CALL X04BFFP(ICNTXT,NOUT,1,NX,A(0,1),LDA,FORMAT,TITOP,CNUMOP,ICOFF,
+              A(N+1,1),LDA,IFAIL)
*
*      Print the eigenvectors on the root processor
*
      IF (ROOT) THEN
        WRITE (NOUT,*) 'Eigenvectors of the matrix'
        WRITE (NOUT,*)
          CNUMOP = 'G'
      END IF
      IFAIL = 0
*
      CALL X04BFFP(ICNTXT,NOUT,N,NX,A(1,1),LDA,FORMAT,TITOP,CNUMOP,ICOFF,
+              A(N+1,1),LDA,IFAIL)
*
*      Print the eigenvectors corresponding to positive eigenvalues
*
      IF (ROOT) THEN
        WRITE (NOUT,*) 'Eigenvectors (positive eigenvalues)'
        WRITE (NOUT,*)
          TITOP = 'N'
          CNUMOP = 'L'
      END IF
*
*      Count positive eigenvalues
*
      JP = 0
      DO 20 J = 1, NX
        IF (A(0,J).GT.ZERO) THEN
          JP = JP + 1
        END IF
      20 CONTINUE
*
*      (Integer function OFFSET gives the additive offset required to
*      transform the local column number to the global column number.)
*
      ICOFF = OFFSET(ICNTXT,N)
      IFAIL = 0
*
      CALL X04BFFP(ICNTXT,NOUT,N,JP,A(1+NX-JP,1),LDA,FORMAT,TITOP,CNUMOP,
+              ICOFF,A(N+1,1),LDA,IFAIL)
*
*      Undefine the 2D processor grid
*
      IFAIL = 0
*
      CALL Z01ABFP(ICNTXT,'N',IFAIL)
*
      STOP
      END
*
      SUBROUTINE GMATA(M,J1,JL,AL,LDAL)

```

```

*
*   GMATA generates the block A( 1: M, J1: JL ) of the
*   matrix A such that
*
*   a(i,j) = max(i,j)+1  if i=j
*   a(i,j) = max(i,j)    otherwise
*
*   in the array AL.
*
*   .. Scalar Arguments ..
*   INTEGER          J1, JL, LDAL, M
*   .. Array Arguments ..
*   DOUBLE PRECISION AL(LDAL,*)
*   .. Local Scalars ..
*   INTEGER          I, J, L
*   .. Intrinsic Functions ..
*   INTRINSIC        MAX
*   .. Executable Statements ..
*   L = 1
*   DO 40 J = J1, JL
*       DO 20 I = 1, M
*           IF (I.NE.J) THEN
*               AL(I,L) = MAX(I,J)
*           ELSE
*               AL(I,L) = MAX(I,J) + 1
*           END IF
*       20 CONTINUE
*       L = L + 1
*   40 CONTINUE
*
*   End of GMATA.
*
*   RETURN
*   END
*
*   INTEGER FUNCTION OFFSET(ICNTXT,N)
*
*   Integer function OFFSET gives the (additive) offset required to
*   transform the local column number to the global column number if
*   the distribution corresponds to the data structure
*   in F01ZRFP, F02FQFP or F02WQFP.
*
*   .. Scalar Arguments ..
*   INTEGER          ICNTXT, N
*   .. Local Scalars ..
*   INTEGER          ME, MP, MYCOL, MYROW, NB, NP, P
*   .. External Subroutines ..
*   EXTERNAL         BLACS_GRIDINFO
*   .. Intrinsic Functions ..
*   INTRINSIC        MOD
*   .. Executable Statements ..
*   CALL BLACS_GRIDINFO(ICNTXT,MP,NP,MYROW,MYCOL)
*   P = MP*NP
*   IF (P.GT.N) THEN
*       NB = 1
*   ELSE
*       NB = N/P
*       IF (MOD(N,P).NE.0) THEN

```

```

        NB = NB + 1
      END IF
    END IF
    ME = MP*MYROW + MYCOL
    OFFSET = ME*NB
*
*   End of OFFSET.
*
    RETURN
  END

```

8.2 Example Data

None.

8.3 Example Results

X04BFFP Example Program Results

Matrix

Array from logical processor 0, 0

1	2	3	4
2.00000	2.00000	3.00000	4.00000
2.00000	3.00000	3.00000	4.00000
3.00000	3.00000	4.00000	4.00000
4.00000	4.00000	4.00000	5.00000
5.00000	5.00000	5.00000	5.00000
6.00000	6.00000	6.00000	6.00000
7.00000	7.00000	7.00000	7.00000
8.00000	8.00000	8.00000	8.00000
9.00000	9.00000	9.00000	9.00000
10.00000	10.00000	10.00000	10.00000
11.00000	11.00000	11.00000	11.00000
12.00000	12.00000	12.00000	12.00000
13.00000	13.00000	13.00000	13.00000
14.00000	14.00000	14.00000	14.00000

Array from logical processor 0, 1

5	6	7	8
5.00000	6.00000	7.00000	8.00000
5.00000	6.00000	7.00000	8.00000
5.00000	6.00000	7.00000	8.00000
5.00000	6.00000	7.00000	8.00000
6.00000	6.00000	7.00000	8.00000
6.00000	7.00000	7.00000	8.00000
7.00000	7.00000	8.00000	8.00000
8.00000	8.00000	8.00000	9.00000
9.00000	9.00000	9.00000	9.00000
10.00000	10.00000	10.00000	10.00000
11.00000	11.00000	11.00000	11.00000
12.00000	12.00000	12.00000	12.00000
13.00000	13.00000	13.00000	13.00000
14.00000	14.00000	14.00000	14.00000

Array from logical processor 1, 0

9	10	11	12
9.00000	10.00000	11.00000	12.00000
9.00000	10.00000	11.00000	12.00000
9.00000	10.00000	11.00000	12.00000
9.00000	10.00000	11.00000	12.00000
9.00000	10.00000	11.00000	12.00000
9.00000	10.00000	11.00000	12.00000
9.00000	10.00000	11.00000	12.00000
9.00000	10.00000	11.00000	12.00000
10.00000	10.00000	11.00000	12.00000
10.00000	11.00000	11.00000	12.00000
11.00000	11.00000	12.00000	12.00000
12.00000	12.00000	12.00000	13.00000
13.00000	13.00000	13.00000	13.00000
14.00000	14.00000	14.00000	14.00000

Array from logical processor 1, 1

13	14
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
13.00000	14.00000
14.00000	14.00000
14.00000	15.00000

Eigenvalues

1	2	3	4
-23.87291	-4.30111	-1.33903	-0.34228
5	6	7	8
0.11123	0.35439	0.49876	0.59037
9	10	11	12
0.65096	0.69185	0.71931	0.73694
13	14		
0.74682	143.75472		

Eigenvectors of the matrix

1	2	3	4
0.39855	-0.37374	0.35925	-0.34208
0.38253	-0.30324	0.20566	-0.08723
0.35113	-0.17553	-0.03585	0.23261
0.30561	-0.01472	-0.26204	0.37915
0.24780	0.14888	-0.37620	0.24323

0.18003	0.28439	-0.32952	-0.07390
0.10502	0.36625	-0.14196	-0.33597
0.02579	0.37902	0.10629	-0.34775
-0.05447	0.32030	0.30910	-0.10045
-0.13255	0.20115	0.37976	0.22168
-0.20530	0.04406	0.28806	0.37866
-0.26979	-0.12134	0.07321	0.25354
-0.32344	-0.26386	-0.17294	-0.06047
-0.36408	-0.35659	-0.34515	-0.32943

5	6	7	8
-0.32113	0.29634	0.26792	0.23619
0.04019	-0.16267	-0.26660	-0.34039
0.35629	-0.36972	-0.26924	-0.08600
0.27151	-0.00410	0.26527	0.37834
-0.11876	0.36786	0.27055	-0.08092
-0.37540	0.17004	-0.26393	-0.34263
-0.20966	-0.29116	-0.27185	0.23209
0.19198	-0.30138	0.26259	0.24023
0.37762	0.15522	0.27315	-0.33808
0.13838	0.37139	-0.26124	-0.09107
-0.25656	0.01231	-0.27444	0.37827
-0.36283	-0.36584	0.25989	-0.07582
-0.06086	-0.17733	0.27572	-0.34481
0.30958	0.28585	-0.25853	0.22795

9	10	11	12
0.20151	0.16432	-0.12507	-0.08426
-0.37581	-0.36892	0.32051	0.23606
0.12356	0.29503	-0.37576	-0.34099
0.26894	0.00157	0.26664	0.37821
-0.35618	-0.29699	-0.04089	-0.34034
0.03915	0.36821	-0.20275	0.23489
0.32232	-0.16148	0.35770	-0.08281
-0.31795	-0.16715	-0.35619	-0.08572
-0.04730	0.36960	0.19887	0.23722
0.35886	-0.29305	0.04543	-0.34163
-0.26310	-0.00472	-0.26986	0.37820
-0.13129	0.29892	0.37625	-0.33969
0.37666	-0.36748	-0.31806	0.23372
-0.19452	0.15863	0.12075	-0.08136

13	14
0.04240	0.21152
-0.12506	0.21300
0.20144	0.21597
-0.26769	0.22046
0.32048	0.22649
-0.35717	0.23411
0.37590	0.24336
-0.37573	0.25433
0.35668	0.26707
-0.31970	0.28168
0.26665	0.29827
-0.20019	0.31695
0.12367	0.33785
-0.04093	0.36111

Eigenvectors (positive eigenvalues)

5	6	7	8
-0.32113	0.29634	0.26792	0.23619
0.04019	-0.16267	-0.26660	-0.34039
0.35629	-0.36972	-0.26924	-0.08600
0.27151	-0.00410	0.26527	0.37834
-0.11876	0.36786	0.27055	-0.08092
-0.37540	0.17004	-0.26393	-0.34263
-0.20966	-0.29116	-0.27185	0.23209
0.19198	-0.30138	0.26259	0.24023
0.37762	0.15522	0.27315	-0.33808
0.13838	0.37139	-0.26124	-0.09107
-0.25656	0.01231	-0.27444	0.37827
-0.36283	-0.36584	0.25989	-0.07582
-0.06086	-0.17733	0.27572	-0.34481
0.30958	0.28585	-0.25853	0.22795
9	10	11	12
0.20151	0.16432	-0.12507	-0.08426
-0.37581	-0.36892	0.32051	0.23606
0.12356	0.29503	-0.37576	-0.34099
0.26894	0.00157	0.26664	0.37821
-0.35618	-0.29699	-0.04089	-0.34034
0.03915	0.36821	-0.20275	0.23489
0.32232	-0.16148	0.35770	-0.08281
-0.31795	-0.16715	-0.35619	-0.08572
-0.04730	0.36960	0.19887	0.23722
0.35886	-0.29305	0.04543	-0.34163
-0.26310	-0.00472	-0.26986	0.37820
-0.13129	0.29892	0.37625	-0.33969
0.37666	-0.36748	-0.31806	0.23372
-0.19452	0.15863	0.12075	-0.08136
13	14		
0.04240	0.21152		
-0.12506	0.21300		
0.20144	0.21597		
-0.26769	0.22046		
0.32048	0.22649		
-0.35717	0.23411		
0.37590	0.24336		
-0.37573	0.25433		
0.35668	0.26707		
-0.31970	0.28168		
0.26665	0.29827		
-0.20019	0.31695		
0.12367	0.33785		
-0.04093	0.36111		